
Video Paging Demo

ZBASIC FLIPDEMO.BAS

I am sure most of you are aware that if your
Z-100's video board is populated with all three
banks of RAM, you have color capability, or grey
scale capability on monochrome displays.

You may also recall that if you were using 32K
RAM chips, you have one page of video memory.
And, if you had 64K chips, you had the capa-
bility of displaying two pages of video memory.

However, not all that much had been written
about how to use this video paging capability.
In fact, it was difficult just finding it
mentioned in all of the Z-100's copious
documentation.

Looking in the LifeLine's "Sextant" Index, I
found one small article that gave the proce-
dures, though they used the "C" programming
language. It was not difficult, however, to
translate into BASIC.

The article was written by William N. Locke in
his "Z-100 Notebook" column of the July-August
1986 issue of "Sextant", #23. In it, he presents
a program called FLIP DEMO. I have included the
ZBASIC version, FLIPDEMO.BAS, below.

Without going into great detail, the second page
of video gives us the capability of fast anima-
tion, by writing on the invisible page of video
memory, then toggling pages to make it visible.
This eliminates the scrolling down the page so
commonly seen when writing graphics directly to
the screen!

The Z-100 uses the 68A45 CRT Controller (CRT-C)
chip to keep track of the video memory and
screen control. This controller chip displays
the green, red, and blue banks of video memory
as individual pixels on the computer's screen.

There are two separate CRT-C ports that we are
concerned with for video screen control:

    Address Port: used for identifying which
particular CRT-C register will be used for
accessing the desired data. This port number is
0DCh (hexadecimal) or 220d (decimal).

    Data Port In: used for passing data to and
from video memory. This port is 0DDh or 221d.

In order to flip the display between the two
pages, we must toggle the fourth bit of register
number 12, the most significant byte of the
Start Read Address Register used by the CRT-C.

First we place a 12 in the Start Read Address
Register, Port 0DCh (220d). Then we toggle this
bit by reading register 12 at Port 0DDh (221d),
exclusive ORing this byte with the number 008h
or 08d (00001000 binary), then writing the byte
back to register 12. See the 3000 code for
FLIPDEMO.BAS on the next page.

The Z-100's 8088 CPU accesses the video RAM
through a chip known as the 68A21 Address Latch,
which uses an internal register to keep track of
video RAM access by the CPU.

This is different from the CRT-C's access
discussed above, which controlled which page of
video memory is displayed on the screen. The
CPU's direct access allows us to change the
information in the hidden page of video RAM,
before the page is displayed!

The Address Latch register is controlled through
port number 0DAh (218d), and CPU access to video
RAM pages may be changed by toggling the most
significant bit in this register. The initial
latch value is read, the value is exclusive ORed
with 080h (128d) (10000000 binary), then the
value is written back to the register. See the
2000 code in FLIPDEMO.BAS.

All of the CRT-C registers and their functions
are described in the Z-100 Technical Manual,
Appendix D, pages 153 and 154.

The Address Latch is described on page 4.39 of
the Z-100 Technical Manual.

Rules for Video Page Flipping:

1. Enable Line 25

2. Turn off the cursor and leave it off
   until you are done flipping.

3. Prior to writing on the back page, clear
   the screen and line 25 on that page.

4. Don't allow either page to scroll, as
   it will mess up the other page.
 
If you have any questions or comments, please
email me at:

z100lifeline@swvagts.com

   ~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

10 REM -------------------  FLIPDEMO.BAS  ---------------------------
20 REM                      by Steven Vagts, Z-100 LifeLine,  12-8-03
30 REM                         211 Sean Way, Hendersonville, NC 28792
40 REM     ZBASIC Program for the H/Z-100   4-8-2022
50 REM --------------------------------------------------------------

100 E$=CHR$(27): REM E$=ESCape
110 ENA25$=E$+"x1": CLR25$=E$+"y1": REM Enable/Disable line 25
120 X$=STRING$(70,88): REM A string of 70 Xs
130 Y$=STRING$(70,89): REM A string of 70 Ys

1000 REM Set Up Display
1010 CLS: REM Clear first page of video
1020 LOCATE ,,0: REM Always turn cursor OFF until done flipping
1030 PRINT ENA25$: REM Line 25 MUST be enabled
1040 PRINT "Please wait while I work on the back page."
1050 GOSUB 2000: REM Toggle CPU access to second page
1060 CLS: REM Clear second page of video

1100 REM Load second video page with characters
1110 PRINT TAB(20);"This is the Second Video Page"
1120 FOR I=1 TO 20: PRINT TAB(5);Y$: NEXT
1130 GOSUB 2000: REM Toggle CPU access back to first video page
1140 REM Always keep track of which page the CPU is on!
1150 CLS: REM Clear first page of wait message.
1160 REM Load first video page with characters
1170 PRINT TAB(20);"This is the First Video Page"
1180 FOR I=1 TO 20: PRINT TAB(5);X$: NEXT

1300 REM Display alternate video pages
1310 FOR I=1 TO 10: REM Flip 10 times
1320 REM To end on the right page, always flip an even number of times
1330 FOR J=1 TO 1000: NEXT J: REM J acts as a delay
1340 GOSUB 3000: REM Toggle display
1350 NEXT I

1400 REM We're done! Clean up
1410 CLS: LOCATE ,,1: REM Clear screen and turn cursor ON
1420 STOP

2000 REM Toggle CPU Access to Video Page
2010 LATCHVAL=INP(218): REM Get initial value of Address Latch Register
2020 LATCHVAL=LATCHVAL XOR 128: REM Exclusive OR with 080h (128d)
2030 OUT 218,LATCHVAL: REM Send new Latch Value byte to Port 218
2040 RETURN

3000 REM Toggle Display
3010 OUT 220,12: REM Place a 12 in the Start Read Address Register, Port 220
3020 STARTADD=INP(221): REM Get current Start Read Address from Port 221
3030 STARTADD=STARTADD XOR 8: REM Exclusive OR with 08h
3040 OUT 221,STARTADD: REM Send new Start Read Address byte to Port 221
3050 RETURN

